

Дата: 2021/11/08
Предмет: Вычислительная математика
Тема: Вычисление обратной матрицы методом Халецкого
Тип занятия: Лекция
Группы: ИВТ-20, ВМК-20

Рассмотрим задачу нахождения обратной к невырожденной квадратной матрице. Дана матрица размера $n \times n$

$$A = \begin{pmatrix} a_{00} & a_{01} & \dots & a_{0,n-1} \\ a_{10} & a_{11} & \dots & a_{1,n-1} \\ \dots & \dots & \dots & \dots \\ a_{n-1,0} & a_{n-1,1} & \dots & a_{n-1,n-1} \end{pmatrix}$$

Необходимо найти к ней обратную (такую квадратную матрицу A^{-1} , что $AA^{-1} = A^{-1}A = E$, где E – единичная матрица, у которой на главной диагонали единицы, а на остальных местах нули).

Этап 1.

Известно, что любую невырожденную матрицу можно представить в виде произведения двух треугольных матриц – нижней треугольной матрицы B и верхней треугольной матрицы T :

$$A = BT = \begin{pmatrix} b_{00} & 0 & \dots & 0 \\ b_{10} & b_{11} & \dots & 0 \\ \dots & \dots & \dots & \dots \\ b_{n-1,0} & b_{n-1,1} & \dots & b_{n-1,n-1} \end{pmatrix} \begin{pmatrix} 1 & t_{01} & t_{02} & \dots & t_{0,n-1} \\ 0 & 1 & t_{12} & \dots & t_{1,n-1} \\ \dots & \dots & \dots & \dots & \dots \\ 0 & 0 & 0 & \dots & 1 \end{pmatrix}$$

Для нахождения элементов матриц B и T применяются явные формулы. Сначала вычисляются коэффициенты первого столбца матрицы B :

$$b_{i0} = a_{i0} \quad (i = 0, \dots, n-1).$$

Затем вычисляются коэффициенты первой строки матрицы T :

$$t_{0j} = \frac{a_{0j}}{b_{00}} \quad (j = 1, \dots, n-1).$$

Кстати, отсюда следует, что должно выполняться условие $a_{00} \neq 0$.

Далее совместно (в одном цикле по параметру $k = 1, \dots, n-1$) вычисляются остальные коэффициенты:

$$b_{ik} = a_{ik} - \sum_{m=0}^{k-1} b_{im}t_{mk}, \quad (i = k, \dots, n-1),$$

$$t_{kj} = \frac{1}{b_{kk}} \left(a_{kj} - \sum_{m=0}^{k-1} b_{km}t_{mj} \right), \quad (j = k+1, \dots, n-1).$$

Вычисление коэффициентов t_{kj} надо завершить при $k = n-2$, это надо учесть при разработке программы.

Этап 2.

После получения матриц B и T на втором этапе вычисляется матрица, обратная к B :

$$B^{-1} = \begin{pmatrix} y_{00} & 0 & \dots & 0 \\ y_{10} & y_{11} & \dots & 0 \\ \dots & \dots & \dots & \dots \\ y_{n-1,0} & y_{n-1,1} & \dots & y_{n-1,n-1} \end{pmatrix}$$

с помощью матричного уравнения $BB^{-1} = E$ (где E — единичная матрица). Аналогичным образом определяется матрица, обратная к T :

$$T^{-1} = \begin{pmatrix} x_{00} & x_{01} & x_{02} & \dots & x_{0,n-1} \\ 0 & x_{11} & x_{12} & \dots & x_{1,n-1} \\ \dots & \dots & \dots & \dots & \dots \\ 0 & 0 & 0 & \dots & x_{n-1,n-1} \end{pmatrix}$$

с помощью матричного уравнения $TT^{-1} = E$.

Матрица B^{-1} , обратная к нижней треугольной матрице, также является нижней треугольной матрицей, а матрица T^{-1} , обратная к верхней треугольной матрице, также является верхней треугольной матрицей.

К примеру, если число строк и столбцов матриц $n = 4$, то мы получаем:

$$BB^{-1} = \begin{pmatrix} b_{00}y_{00} & 0 & 0 & 0 \\ b_{10}y_{00} + b_{11}y_{10} & b_{11}y_{11} & 0 & 0 \\ b_{20}y_{00} + b_{21}y_{10} + b_{22}y_{20} & b_{21}y_{11} + b_{22}y_{21} & b_{22}y_{22} & 0 \\ b_{30}y_{00} + b_{31}y_{10} + b_{32}y_{20} + b_{33}y_{30} & b_{31}y_{11} + b_{32}y_{21} + b_{33}y_{31} & b_{32}y_{22} + b_{33}y_{32} & b_{33}y_{33} \end{pmatrix}$$

Делая обобщение на произвольную размерность, получаем формулы для вычисления элементов матрицы B^{-1} . Если i — номер строки этой матрицы, а j — номер столбца (i изменяется от 0 до $n - 1$ и j изменяется от 0 до $n - 1$), то:

- если $j > i$, то $y_{ij} = 0$;
- если $j = i$, то $y_{ij} = \frac{1}{b_{ii}}$;
- если $j < i$, то $y_{ij} = -\frac{1}{b_{ii}} \cdot \sum_{m=j}^{i-1} b_{im}y_{mj}$.

Аналогично, рассматривая пример при $n = 4$, имеем:

$$TT^{-1} = \begin{pmatrix} x_{00} & x_{01} + t_{01}x_{11} & x_{02} + t_{01}x_{12} + t_{02}x_{22} & x_{03} + t_{01}x_{13} + t_{02}x_{23} + t_{03}x_{33} \\ 0 & x_{11} & x_{12} + t_{12}x_{22} & x_{13} + t_{12}x_{23} + t_{13}x_{33} \\ 0 & 0 & x_{22} & x_{23} + t_{23}x_{33} \\ 0 & 0 & 0 & x_{33} \end{pmatrix}$$

Делая обобщение на произвольную размерность, получаем формулы для вычисления элементов матрицы T^{-1} . Если i — номер строки этой матрицы, а j — номер столбца (i изменяется от $n - 1$ до 0 с шагом -1 и j изменяется от $n - 1$ до 0 с шагом -1), то:

- если $j < i$, то $x_{ij} = 0$;
- если $j = i$, то $x_{ij} = 1$;
- если $j > i$, то $x_{ij} = - \sum_{m=i+1}^j t_{im} x_{mj}$.

Этап 3.

На третьем этапе вычисляется матрица, обратная к A , по формуле

$$A^{-1} = T^{-1} B^{-1}$$

Если i – номер строки этой матрицы, а j – номер столбца (i изменяется от 0 до $n - 1$ и j изменяется от 0 до $n - 1$), то:

$$a_{ij}^{-1} = \sum_{m=0}^{n-1} x_{im} y_{mj}.$$

Код 1. Вычисление обратной матрицы методом Халецкого. Исходная матрица A находится в диапазоне ячеек B1:G6

```
Public Sub Main()

    'Определение массивов
    Const n = 6
    Dim a(n, n) As Double
    Dim b(n, n) As Double
    Dim t(n, n) As Double
    Dim y(n, n) As Double
    Dim x(n, n) As Double
    Dim o(n, n) As Double

    'Первичное заполнение матриц A, B, T
    For i = 0 To n - 1
        For j = 0 To n - 1
            a(i, j) = Cells(i + 1, j + 2)
            b(i, j) = 0
            If i = j Then
                t(i, j) = 1
            Else
                t(i, j) = 0
            End If
        Next j
    Next i

    'Вычисление матриц B и T
    For i = 0 To n - 1
        b(i, 0) = a(i, 0)
    Next i

    For j = 1 To n - 1
        t(0, j) = a(0, j) / b(0, 0)
    Next j

    For k = 1 To n - 1
        For i = k To n - 1
            b(i, k) = a(i, k)
            For m = 0 To k - 1
                b(i, k) = b(i, k) - b(i, m) * t(m, k)
            Next m
        Next i
        If k <= n - 2 Then
```

```

        For j = k + 1 To n - 1
            t(k, j) = a(k, j)
            For m = 0 To k - 1
                t(k, j) = t(k, j) - b(k, m) * t(m, j)
            Next m
            t(k, j) = t(k, j) / b(k, k)
        Next j
    End If
Next k

```

```

'Вывод матриц В и Т
For i = 0 To n - 1
    For j = 0 To n - 1
        Cells(i + 8, j + 2) = b(i, j)
        Cells(i + 15, j + 2) = t(i, j)
    Next j
Next i

```

```

'Вычисление и вывод матрицы Y
For i = 0 To n - 1
    For j = 0 To n - 1
        If j > i Then
            y(i, j) = 0
        ElseIf j = i Then
            y(i, j) = 1 / b(i, i)
        ElseIf j < i Then
            y(i, j) = 0
            For m = j To i - 1
                y(i, j) = y(i, j) - b(i, m) * y(m, j)
            Next m
            y(i, j) = y(i, j) / b(i, i)
        End If
        Cells(i + 22, j + 2) = y(i, j)
    Next j
Next i

```

```

'Вычисление и вывод матрицы X
For i = n - 1 To 0 Step -1
    For j = n - 1 To 0 Step -1
        If j < i Then
            x(i, j) = 0
        ElseIf j = i Then
            x(i, j) = 1
        ElseIf j > i Then
            x(i, j) = 0
            For m = i + 1 To j
                x(i, j) = x(i, j) - t(i, m) * x(m, j)
            Next m
        End If
        Cells(i + 29, j + 2) = x(i, j)
    Next j
Next i

```

```

'Вычисление и вывод матрицы, обратной к А
For i = 0 To n - 1
    For j = 0 To n - 1
        o(i, j) = 0
        For m = 0 To n - 1
            o(i, j) = o(i, j) + x(i, m) * y(m, j)
        Next m
        Cells(i + 36, j + 2) = o(i, j)
    Next j
Next i

```

End Sub
